

**SAS code for “Mixed-Effects Regression Models with Heterogeneous Variance:  
Analyzing Ecological Momentary Assessment (EMA) Data of Smoking”**

Donald Hedeker & Robin Mermelstein

In T.D. Little, J.A. Bovaird, & N.A. Card (Eds.), *Modeling Contextual Effects in Longitudinal Studies* (pp. 183-206). Erlbaum, Mahwah, NJ, 2007.

**SAS syntax for models in TABLE 8.3** Here, SMKLEVEL is coded 0, 1, 2 for the three smoking history groups, and is used as a categorical variable in most of the syntax below. The variable SMKLEV is set equal to SMKLEVEL for use as a linear variable in Model Ib, *i.e.*,

```
SMKLEV = SMKLEVEL;
```

For Model IIc, SMKLEVR is a reverse-coded version of SMKLEV, *i.e.*,

```
IF SMKLEV = 0 THEN SMKLEVR = 2;  
IF SMKLEV = 1 THEN SMKLEVR = 1;  
IF SMKLEV = 2 THEN SMKLEVR = 0;
```

Model Ia

```
PROC MIXED METHOD=REML COVTEST;  
    CLASS ID SMKLEVEL;  
    MODEL CPHYS = SMKLEVEL /SOLUTION;  
    RANDOM INTERCEPT /SUB=ID;
```

Model Ib

```
PROC MIXED METHOD=REML COVTEST;  
    CLASS ID SMKLEVEL;  
    MODEL CPHYS = SMKLEVEL /SOLUTION;  
    RANDOM INTERCEPT /SUB=ID;  
    REPEATED /SUB=ID LOCAL=EXP(SMKLEV);
```

Model IIc

```
PROC MIXED METHOD=REML COVTEST;  
    CLASS ID SMKLEVEL;  
    MODEL CPHYS = SMKLEVEL /SOLUTION;  
    RANDOM INTERCEPT SMKLEVR /SUB=ID TYPE=VC;  
    REPEATED /SUB=ID GROUP=SMKLEVEL;
```

Model IIIc

```
PROC MIXED METHOD=REML COVTEST;  
    CLASS ID SMKLEVEL;  
    MODEL CPHYS = SMKLEVEL /SOLUTION;  
    RANDOM INTERCEPT /SUB=ID GROUP=SMKLEVEL;  
    REPEATED /SUB=ID GROUP=SMKLEVEL;
```

**SAS syntax for models in TABLE 8.6** Here, GROW2GP is coded 0, 1 for the two smoking growth groups, and is used as a linear variable in most of the syntax below. The variable GROW2GPC is set equal to GROW2GP for use as a categorical variable in the second model below, *i.e.*,

```
GROW2GPC = GROW2GP;
```

Model BS(T) WS(GT)

```
PROC MIXED METHOD=REML COVTEST;  
    CLASS ID;  
    MODEL CPHYS = GROW2GP TIME GROW2GP*TIME /SOLUTION;  
    RANDOM INTERCEPT TIME /SUB=ID TYPE=UN;  
    REPEATED /SUB=ID LOCAL=EXP(GROW2GP TIME GROW2GP*TIME);
```

Model BS(GT) WS(GT)

```
PROC MIXED METHOD=REML COVTEST;  
    CLASS ID GROW2GPC;  
    MODEL CPHYS = GROW2GP TIME GROW2GP*TIME /SOLUTION;  
    RANDOM INTERCEPT TIME /SUB=ID TYPE=UN GROUP=GROW2GPC;  
    REPEATED /SUB=ID LOCAL=EXP(GROW2GP TIME GROW2GP*TIME);
```